Title

A 709/7090 Fortran II Program to Compute the Neutron-Detection Efficiency of Plastic Scintillator for Neutron Energies from 1 to 300 Mev

Author

Kurz, Richard J.

Publication Date

1964-03-18

UCRL-11339 c.2
UC-34 Physics
TID-4500 (24th Ed.)

UNIVERSITY OF CALIFORNIA
Lawrence Radiation Laboratory
Berkeley, California

AEC Contract No. W-7405-eng-48

A 709/7090 FORTRAN II PROGRAM TO COMPUTE THE NEUTRON-DETECTION EFFICIENCY OF PLASTIC SCINTILLATOR FOR NEUTRON ENERGIES FROM 1 TO 300 MeV

Richard J. Kurz

March 18, 1964

UNIVERSITY OF CALIFORNIA
Lawrence Radiation Laboratory
Berkeley, California

AEC Contract No. W-7405-eng-48

February 16, 1965

ERRATA

TO: All recipients of UCRL-11339 UC-34 Physics

FROM: Technical Information Division

Subject: UCRL-11339, "A 709/7090 Fortran II Program to Compute the Neutron-Detection Efficiency of Plastic Scintillator for Neutron Energies from 1 to 300 MeV."

Please make the following corrections on subject report.

Page 3, Eq. (1): change to

$$\epsilon(E) = C \int_0^\infty dT \exp\{-1/2[(T-T_0)/\tau\ T_0]^2\}\epsilon(E,T), \qquad (1)$$

with

$$C^{-1} = \int_0^\infty dT \exp\{-1/2[(T-T_0)/\tau\ T_0]^2\},$$

Page 6, Table II:

$$W(E,Z) = 1-e^{-\nu\bar{\sigma}(E)Z}/\bar{\sigma}(E)$$

Page 6, Eq. (4):

$$\epsilon_H^1(E,T) = \nu \int_0^\ell e^{-\nu\bar{\sigma}(E)Z} F_1(E,T)dZ \qquad (4)$$

Page 7, Eq. (5):

$$F_1(E,T) = 2\pi \int_{-1}^{X'max} \frac{d\sigma_{n-p}(E,X')}{d\Omega'} dX', \qquad (5)$$

Page 7, Eq. (7):

$$F_2(E,T) = 2\pi \int_{X'max}^{1} \frac{d\sigma_{n-p}}{d\Omega'} \epsilon^1(X')dX', \qquad (7)$$

Page 8, Eq. (11): change to

$$g_a(E, T) = 1/2 \int_{E_a^{min}}^{E_a^{max}} \frac{dE_a}{\bar{\beta}\bar{\gamma}p'} = (E_a^{max} - E_a^{min})/2\bar{\beta}\bar{\gamma}p' , \qquad (11)$$

with the restriction that

$0 \leq g_a(E, T) \leq 1,$

Page 8, Eq. (12); should be followed by:

$$\text{with } C^{-1} = \int_0^1 N(\delta)\, d\delta,$$

Page 9, Eq. (17):

$$f_p(E, T, \delta_p) = 1/2 \int_{X_p'^{min}}^{1} N(\delta_p)\, dX_p' \qquad (17)$$

$$= N(\delta_p)(1 - X_p'^{min})/2,$$

Page 10, Eq. (18), the three symbols $X_p'^{max}$ should be replaced by $X_p'^{min}$, i.e.:

where $X_p'^{min}(E, T, \delta_n)$ is determined by

$$X_p'^{min} = \text{————————} \qquad (18)$$

with the restriction

$-1 \leq X_p'^{min} \leq 1.$

Page 10, Eq. (21);

$$G_2(E, T) = \sigma_2(E) \int_0^1 d\delta_n N(\delta_n) \frac{[1 - f_{3a}(Q, T)]}{2} \int_{-1}^{1} dX' \epsilon^1(X', \delta_n), \qquad (21)$$

Page 10, Eq. after Eq. (22):

$$E_n(E, \delta_n, X') = \overline{\gamma}(\delta_n E_n'^{max} + M_n) + X'\overline{\beta}\overline{\gamma}\,[\delta_n E_n'^{max}(\delta_n E_n'^{max} + 2M_n)]^{1/2} - M_n,$$

Page 10, Eq. (23):

$$T \leq L_{3\alpha}^{min} = 0.046Q + \frac{0.007}{3} Q^2, \text{ i.e. } f_{3\alpha}(Q, T) = 1, \qquad (23)$$

Page 11, Eq. (24):

$$T \geq L_{3\alpha}^{max} = 0.046Q + \frac{0.007}{2} Q^2, \text{ i.e. } f_{3\alpha}(Q, T) = 0, \qquad (24)$$

Page 11, Eq. (25):

$$L_{3\alpha}^{max} \geq T \geq L_{3\alpha}^{min}, \text{ i.e. } 0 \leq f_{3\alpha}(Q, T) \leq 1. \qquad (25)$$

Page 11, Eq. (27):

$$\epsilon_{e\ell} = 2\pi W(E,\ell) \ldots . \qquad (27)$$

Page 11, Eq. (28):

$$\epsilon_{\gamma} = 2\pi W(E,\ell) \ldots . \qquad (28)$$

Page 19: After statement 304, add the following statement:

```
AY(J) = 0.0;
```

After statement 307, add the following statement:

```
AY(J) = 0.5.
```

After C EQUATION (25), statement 309 should be replaced by:

```
309 AY(J) = 3. * AL(J)/Y2 - 1.
    AL(J) = (AL(J) - Y2/3.)*.5
```

Page 20: Statement 531 should be changed to:

```
531 FR(4) = FR(4) + Z(J) * G2 * AY(J)
```

Also, the variable AY(16) should be added to the list of variables in the dimension statement on page 18.

A 709/7090 FORTRAN II PROGRAM TO COMPUTE THE NEUTRON-DETECTION EFFICIENCY OF PLASTIC SCINTILLATOR FOR NEUTRON ENERGIES FROM 1 TO 300 MeV

Contents

A 709/7090 FORTRAN II PROGRAM TO COMPUTE THE NEUTRON-DETECTION EFFICIENCY OF PLASTIC SCINTILLATOR FOR NEUTRON ENERGIES FROM 1 TO 300 MeV

Richard J. Kurz

Lawrence Radiation Laboratory
University of California
Berkeley, California

March 18, 1964

ABSTRACT

A computer program to calculate the neutron-detection efficiency of plastic scintillator for neutron energies from 1 to 300 MeV is described. All known systematics for neutron-carbon and neutron-proton interactions were used in this analysis. The calculation includes first- and second-scattering contributions from interactions with both proton and carbon, saturation effects in the production of scintillation light, and finite resolution of the detector threshold. This report includes (a) the results of sample calculations, (b) comparison with experimental measurements, (c) source-program listings, and (d) cross-section data that are used in the calculation.

I. INTRODUCTION

In scintillator, neutrons are detected by the scintillation light produced by the charged products of interactions between the incident neutrons and the nuclei of the scintillator. One of the main problems involved in the use of a scintillation counter for neutron detection is the determination of its detection efficiency. A computer program to calculate this efficiency in plastic scintillator (composition CH, density 1.05 g/cm^3) for neutron energies from 1 MeV to 300 MeV is described here.

For neutron energies below 10 MeV, the efficiency may be calculated reliably because only neutron-proton (n-p) interactions contribute.[1] Above this energy, interactions with the carbon nuclei become significant. Although considerable information is known for the n-p interactions in the energy range under consideration, almost no systematics, other than the total cross section and elastic cross section, are available for neutron-carbon (n-C) interactions. The n-C reactions that are considered are listed in Table I.

Table I. Neutron-carbon reactions and parameters.

Reaction	Q (MeV)	Threshold (MeV)
C(n, n)C	0	0
$C(n, n'\gamma)C_{4.43}$	- 4.43	4.7
C(n, α)Be	- 5.71	6.2
C(n, n'3α)	- 7.26	7.9
C(n, p)B	-12.59	13.6

For detectors that have dimensions of the order of magnitude of the mean-free path for neutron interactions, contributions to the efficiency from successive scatterings are significant. Rescattering of the final-state neutrons from n-p interactions and the (n, n'3α) reaction is included in this calculation.

Before the calculation is described, a summary of the cross-section data used is presented.

II. CROSS-SECTION DATA

Figure 1 presents the total cross sections for n-p and n-C interactions as a function of neutron energy. The values below 100 MeV are taken from the compilation by Hughes and Schwartz.[2] Above 100 MeV the existing measurements are indicated.[3,4] Also shown are (a) the n-C total nonelastic cross section and (b) the n-C total cross section minus the elastic scattering inside the first diffraction minimum. This latter cross section is used in the calculation and was estimated by the use of both n-C and p-C elastic-scattering data, since n-C and p-C elastic scattering are essentially identical above 14-MeV incident energy.[5]



Fig. 1. Neutron cross-section data.

O: n-p total cross section
●: n-C total cross section
△: n-C total less n-C diffraction elastic cross section
▲: n-C nonelastic cross section

Figure 2 shows the breakdown, according to reaction, of the n-C nonelastic cross section used in the calculation. These values were determined in the following manner. First, the (n, n'γ) cross section was estimated by use of both n-C data and p-C data above 14 MeV. This cross section was subtracted from the total nonelastic cross section. Next the (n, n'3α) reaction cross section was estimated from the measurements between 10 and 20 MeV[6] and data at 90 MeV.[7] The data at 90 MeV were taken in a cloud chamber and indicate a high multiplicity of reactions that were grouped as two-prong and more-than-two-prong reactions. The data for more-than-two-prong reactions were used as the 90-MeV value of the (n, n'3α) reaction cross section. At this incident energy the details of the reaction are not critical to the calculation. The principal information is that all n-C nonelastic interactions produce charged secondaries. This estimate of the (n, n'3α) cross section was subtracted and the remaining cross section was treated as that for (n, α) and (n, p) in a manner suggested by the shape of the curve and the location of the thresholds for these reactions. Actually the C(n, d)B reaction has a threshold at 14.9 MeV, so that the treatment of what will be called the (n, p) reaction includes both. The two-prong category of the 90-MeV data was used as the value of the (n, p) cross section at that energy.

The n-p angular-distribution data used in the computation were from the parameterization by Clements and Winsberg.[8] If these angular distributions are integrated to yield a total n-p cross section, there exist some discrepancies with respect to the measured total n-p cross section. Appropriate normalization corrections are included in the program.

III. EFFICIENCY CALCULATION

The efficiency program TOTEFF calculates for a specified series of neutron energies, E, the total efficiency, $\epsilon(E)$. This quantity is the result of a folding integration over the detection threshold, T, of the type

$$\epsilon(E) = C\int_0^\infty dT \exp\left\{-\tfrac{1}{2}\left[(T-T_0)/\tau T_0\right]^2\right\} \epsilon(E, T), \qquad C^{-1} = \int_0^\infty dT\, e^{-\frac{1}{2}\left[(T-T_0)/\tau T_0\right]^2} \quad (1)$$

where T_0 is the mean threshold, τ is the fractional resolution, and $\epsilon(E, T)$ is the efficiency at energy E and threshold T. This function, $\epsilon(E, T)$, is in turn computed in two portions: $\epsilon_H(E, T)$, the efficiency for an initial interaction with hydrogen, and $\epsilon_C(E, T)$, the efficiency for an initial interaction with a carbon nucleus. The essential elements of the program used to compute these quantities are two subroutines. One, HYD(E, T), computes the single-scattering efficiency for n-p collisions; the other, CARB(E, T), computes the single-scattering efficiency for n-C collisions from the (n, α), (n, n'3α), and (n, p) reactions. Since they contribute only through rescattering, the (n, n) and (n, n'γ) reactions are treated separately.

Multiple-scattering effects are calculated by integrating over the neutrons in the final state of primary reactions that did not produce sufficient light to be detected. This integration involves the successive use of HYD(E', T') and CARB(E', T'), where E' is the scattered neutron energy



Fig. 2. Breakdown of n-C nonelastic cross section.

Below 8 MeV all nonelastic cross section taken as (n, n' γ)

A: (n, α) threshold
B: (n, n' 3α) threshold
C: (n, p) threshold
D: (n, d) threshold

O: nonelastic
▲: (n, n'γ)
△: (p, p'γ)
□: (n, n' 3α)
●: (n, α)
■: value for the more-than-two-prong reactions from Ref. 7.

and T' is the light threshold reduced by the light produced in the primary reaction. This integration involves weighting these efficiencies by the differential distributions of the first reaction and the mean escape distance at the angle of the primary scattering. The use of an escape distance averaged over the counter geometry is an approximation that is reasonable as long as the dimensions of the counter are not large compared with the mean free path of the incident neutrons. It is necessary to include a subroutine that computes this mean distance for escaping the detector as a function of the angle of scattering. The geometrical treatment used in a sample efficiency calculation is presented in Appendix C.

In terms of particle energy the detection threshold is a function of the type of particle. Because of saturation effects in the scintillation process, heavily ionizing particles are less efficient in producing scintillation light. These saturation effects must be taken into account when considering the light output of the various interactions in the detector. The response of plastic scintillator calculated and measured by Gooding and Pugh[9] is used in this calculation. Their data were empirically parameterized in two ways. First, the light output of a particle as a function of its energy in units of the energy of an electron required to produce the same light output is given by

$$L_p(E_p) = 10.68\,[1.0 - \exp(-0.07\,E_p^{0.89})] + 0.929\,E_p$$

for protons, and

$$L_\alpha(E_\alpha) = 0.046\,E_\alpha + 0.007\,E_\alpha^2 \qquad (2)$$

for alphas in the range

$$0 \leq E_\alpha \leq 25 \text{ MeV}.$$

Second, the inverse relationships that give the particle energy as a function of its light output, T, are

$$E_p(T) = 11.5\,[1.0 - \exp(-0.2T^{0.62})] + 1.077\,T \qquad (3)$$

for protons, and

$$E_\alpha(T) = 1.5T + 50.0\,[1.0 - \exp(-0.16T^{0.62})]$$

for alpha particles.

The details of the calculation of $\epsilon_H(E, T)$ and $\epsilon_C(E, T)$ are presented in the next two sections. Table II lists and defines symbols used in these sections. The results of some sample calculations are presented in a third section.

Table II. Definition of symbols (prime refers to c.m. system

E	=	incident neutron energy
T	=	scintillation light-detection threshold in units of equivalent electron energy
θ, θ'	=	angle with respect to the incident neutron direction
X, X'	=	$\cos\theta$, $\cos\theta'$
E_i, E_i'	=	energy of particle i
$\bar{\beta}$	=	velocity of the center-of-mass system with respect to the laboratory system
$\bar{\gamma}$	=	$(1 - \bar{\beta}^2)^{-1/2}$
p'	=	momentum
M_i	=	rest mass of particle i
$\bar{\sigma}$	=	sum of the neutron-proton and neutron-carbon total cross sections
E^*	=	total energy (c.m.)
$d\sigma_i/d\Omega'$	=	differential cross section as a function of X'
ν	=	number of hydrogen or carbon nuclei per cm^3 in the scintillator
$d(\theta)$	=	mean escape distance at angle (θ)
Z	=	distance along the incident-neutron direction
$W(E, Z)$	=	$2\pi[1 - \exp(-\nu\bar{\sigma}(E)Z)]/\bar{\sigma}(E)$

A. Hydrogen Interactions

The hydrogen single-scattering efficiency for a length ℓ scintillator is given by

$$\epsilon_H{}^1(E, T) = 2\pi\nu \int_0^\ell e^{-\nu\bar{\sigma}(E)Z} F_1(E, T)dZ \qquad (4)$$

$$= W(E, \ell) F_1(E, T),$$

where $F_1(E, T)$ is the portion of the n-p cross section that produces detectable recoil protons. HYD(E, T) computes $F_1(E, T)$ according to the relation

$$F_1(E, T) = \int_{-1}^{X'^{max}} \frac{d\sigma_{n-p}(E, X')}{d\Omega'} dX', \tag{5}$$

where X'^{max} is the value of X' such that the recoil proton produces a light output in the scintillator equal to T. In other words,

$$E_p = \bar{\gamma}(E_p' + M_p - \bar{\beta} p' X'^{max}) - M_p = E_p(T),$$

where $E_p(T)$ is given by Eq. (3), and $\bar{\gamma}$, $\bar{\beta}$, E_p', and p' are functions of E. Then

$$X'^{max} = \{\bar{\gamma}(E_p' + M_p) - [E_p(T) + M_p]\}/\bar{\beta}\bar{\gamma}p'. \tag{6}$$

Rescattering of the neutrons leaving the primary interaction with hydrogen is included by calculating

$$F_2(E, T) = \int_{X'^{max}}^{1} \frac{d\sigma_{n-p}}{d\Omega'} \epsilon^1(X') dX', \tag{7}$$

where $\epsilon^1(X')$ is the total single-scattering efficiency from hydrogen and carbon for a thickness of scintillator equal to the mean escape distance in a direction θ corresponding to X', and where

$$\epsilon^1 = \epsilon_H^1 \{E_n(X'), T - L_p[E_p(X')]\} + \epsilon_C^1 [E_n, T - L_p], \tag{8}$$

and ϵ_H^1 is given by Eq. (4) with ℓ replaced by d(X),

$$X = [(X' + 1)/2]^{1/2},$$

$$E_n(X') = 2M_n \bar{\beta}^2 X'^2/(1 - \bar{\beta}^2 X'^2),$$

$$E_p(X') = M_p\bar{\beta}^2(1 - X')/[1 - \bar{\beta}^2(1 - X')/2)].$$

The quantity ϵ_C^1 is a function for carbon interactions analogous to ϵ_H^1, and is discussed in the next section.

Finally,

$$\epsilon_H(E, T) = W(E, \ell) [F_1(E, T) + F_2(E, T)] . \tag{9}$$

B. Carbon Interactions

The treatments of the kinematics for the three n-C interactions are different. It is assumed that there is no c. m. angular dependence in any of the reactions.

The (n, α) reaction is treated as a two-body reaction. Be^{*9} excited states are not considered because the assumed cross section has become negligible before energies sufficient to allow much excitation have been reached. To determine the fraction of (n, α) interactions detected, it is necessary to use the quantity

$$\frac{d\sigma}{dE_\alpha} = \frac{d\sigma}{d\Omega'} \frac{d\Omega'}{dE_\alpha} = \sigma_T / 2\bar{\beta}\bar{\gamma}p' . \tag{10}$$

For this reaction, CARB(E, T) computes the total fraction of (n, α) reactions detected:

$$g_\alpha(E, T) = \int_{E_\alpha^{min}}^{E_\alpha^{max}} \frac{dE_\alpha}{\bar{\beta}\bar{\gamma}p'} = (E_\alpha^{max} - E_\alpha^{min})/\bar{\beta}\bar{\gamma}p' , \tag{11}$$

where

$$E_\alpha^{max} = \bar{\gamma} E_\alpha' + \bar{\beta}\bar{\gamma}p' + M_\alpha(\bar{\gamma} - 1) ,$$

and

$$E_\alpha^{min} = E_\alpha(T) ,$$

where $E_\alpha(T)$ is as defined in Eq. (3).

The selection of the method for calculating the (n, n'3α) reaction was based on the available experimental data.[6] Although the reaction appears to proceed through intermediate states, the essential features needed here are adequately described by the assumption of a four-body phase-space distribution of the final-state energies. This permits calculation of the average light produced by the three alphas.

The calculation of the fraction of (n, n'3α) reactions detected proceeds in two steps. The four-body phase-space distribution is given by[6]

$$N(\delta)\, d\delta = C\delta^{1/2}(1 - \delta)^2\, d\delta , \tag{12}$$

where δ is the ratio of the kinetic energy of any particle to its maximum possible kinetic energy, in the overall c. m. system. The excitation energy of the recoiling carbon nucleus available to the three alphas, $Q(E, \delta_n)$, is derived from

$$\delta_n = \frac{E_n'}{E_n'^{max}} = \frac{(E^* - M_n)^2 - (M_{C12} + 7.26 + Q)^2}{(E^* - M_n)^2 - (M_{C12} + 7.26)^2} , \tag{13}$$

where E^* is a function of E, or

$$Q = -(M_{C12} + 7.26) + \left[\delta_n (M_{C12} + 7.26)^2 + (1 - \delta_n)(E^* - M_n)\right]^{1/2} .$$

Using Eq. (2), one finds the light produced by the three alphas:

$$L_{3\alpha} = 0.046 \sum_{i=1}^{3} E_{\alpha i} + 0.007 \sum_{i=1}^{3} (E_{\alpha i}^2) \qquad (14)$$

$$= 0.046\, Q + 0.007 \sum_{i} (E_{\alpha i}^2) .$$

For a phase-space distribution of the $E_{\alpha i}$'s, the average fraction of events detected for threshold T is

$$f_{3\alpha}(Q, T) = 3\left[1 - \frac{2\left(\frac{T}{Q} - 0.046\right)}{0.007\, Q}\right], \qquad (15)$$

with the additional restriction that for $f_{3\alpha}(Q, T) \leq 0$, $f_{3\alpha}(Q, T) = 0$, and for $f_{3\alpha}(Q, T) \geq 1$, $f_{3\alpha}(Q, T) = 1$.

For the (n, n'3α) reaction, CARB(E, T) computes the average total fraction detected,

$$g_{3\alpha}(E, T) = \int_0^1 d\,\delta_n\, N(\delta_n)\, f_{3\alpha}\left[Q(\delta_n), T\right] . \qquad (16)$$

The (n, p) reaction was not considered as a two-body reaction. The B^{12} system has a high density of low-lying excited states[10] and the interaction must be of a direct charge-exchange type rather than optical in nature. In the absence of any data, the final-state proton was assumed to have the same energy distribution as the neutron in the (n, n'3α) case. A distribution calculated on the basis of a direct-interaction model might be more accurate, but the added sophistication was not deemed appropriate here in consideration of the speculative nature of the estimate of the (n, p) reaction cross section. The calculation is similar to the (n, α) reaction treatment except that the variable excitation energy of the B^{*12} must be considered. The detected fraction is calculated by integration of the differential distribution in energy and angle,

$$f_p(E, T, \delta_p) = 2\pi \int_{-1}^{X_p'^{,max}} \frac{d^2\sigma}{d\delta_p d\Omega'}\, d\delta_p dX' \qquad (17)$$

$$= C\, \delta_p^{1/2} (1 - \delta_p)^2 (X_p'^{,max} + 1) ,$$

where $X'^{max}(E, T, \delta_p)$ is determined by

$$X_p'^{max} = \frac{M_p + E_p(T) - \bar{\gamma}(\delta_p E_p'^{max} + M_p)}{\bar{\beta}\bar{\gamma}\,[\delta_p E_p'^{max}(\delta_p E_p'^{max} + 2M_p)]^{1/2}}, \tag{18}$$

with the restriction

$$-1 \leq X_p'^{max} \leq 1 .$$

CARB(E, T) calculates the total fraction of (n, p) reactions detected:

$$g_p(E, T) = \int_0^1 d\,\delta_p\, f_p(E, T, \delta_p) . \tag{19}$$

The single-scattering efficiency for n-C interactions is given by

$$\epsilon_C^{\,1}(E, T) = W(E, \ell)\, G_1(E, T) , \tag{20}$$

where

$$G_1(E, T) = \sum_k \sigma_k(E)\, g_k(E, T) ,$$

with the summation performed over the three reactions.

The efficiency derived from the rescattering of the final-state neutrons in (n, n'3α) reactions is calculated according to the expression

$$G_2(E, T) = \int_0^1 d\,\delta_n\, N(\delta_n) \int_0^1 dX'\, \epsilon^1(X', \delta_n) , \tag{21}$$

where $\epsilon^1(X', \delta_n)$ is the single-scattering efficiency for n-p and n-C interactions at the energy of the scattered neutron in the average escape distance in the laboratory system corresponding to X'. Explicitly,

$$\epsilon^1(X', \delta_n) = W[E_n, d(X')]\,[F_1(E_n, T') + G_1(E_n, T')] , \tag{22}$$

where

$$E_n(E, \delta_n, X') = \bar{\gamma}(\delta_n E_n'^{max} + M_n) + \bar{\beta}\bar{\gamma}(\delta_n E_n'^{max} + 2M_n)^{1/2} - M_n ,$$

and T' is the threshold adjusted for the average amount of light produced by the three alphas in the primary (n, n'3α) reaction. The alpha light is a function of Q, which is in turn a function of δ_n [Eq. (13)]. The rescattering calculation is divided into three categories. For

$$T \leq L_{3\alpha}^{min} = 0.046\, Q + \frac{0.007}{3} Q^2 , \tag{23}$$

all first scatterings are detected and no rescattering effects are present. For

$$T \geq L_{3a}^{max} = 0.046\ Q + \frac{0.007}{2}\ Q^2 , \tag{24}$$

no first scatterings are detected and the average threshold for the rescattering calculation is

$$T' = T - \left(0.046\ Q + \frac{0.035}{12}\ Q^2\right).$$

Finally, for

$$L_{3a}^{max} \geq T \geq L_{3a}^{min} ,$$

$$T' = \left[T - \left(0.046\ Q + \frac{0.007}{3}\ Q^2\right)\right]/2 . \tag{25}$$

The total efficiency derived from inelastic n-C reactions with charged products is

$$\epsilon_C(E, T) = W(E, \ell)\ [G_1(E, T) + G_2(E, T)] . \tag{26}$$

The calculation of the contributions to the detection efficiency of the rescattering effects in (n, n) and (n, n'γ) reactions is not performed by the program. At neutron energies greater than 20 MeV, the extreme forward peaking of the (n, n) angular distribution allows a great simplification in the treatment. Since the energy and direction of the scattered neutron differ from those of the incident neutron by small amounts, it is reasonable to neglect this contribution to the n-C total cross section used in the evaluation of $\overline{\sigma}(E)$ in the expressions above. In order to estimate the contribution below 20 MeV, a hand calculation can be performed at 14 MeV based on the data available for $d\sigma(n, n)/d\Omega$.[5] The contribution is given by the expression

$$\epsilon_{e\ell} = W(E, \ell) \int_{-1}^{1} dX\ \frac{d\sigma_{(n, n)}}{d\Omega}\ \epsilon[E_n(X), T]\ \frac{d(X)}{\ell} , \tag{27}$$

where $\epsilon[E_n(X), T]$ is the total efficiency computed by the program.

Both the final-state neutron and the gamma may be detected in the (n, n'γ) reaction. The detection of the 4.43-MeV gamma is principally by Compton scattering. The angular distributions of both the gamma and the neutron have been measured at 14 MeV.[5] The gamma efficiency is given by

$$\epsilon_\gamma = W(E, \ell)\ f(t)\ \rho\ \mu_{eff} \int_{-1}^{1} dX\ \frac{d\sigma_{(n, n'\gamma)}}{d\Omega}\ d(X) , \tag{28}$$

where

$$\mu_{eff} = 7\ \mu_{Comp}^{H}/13 ,$$

μ^{H}_{Comp} is the total Compton attenuation coefficient for hydrogen,[11] $f(t)$ is the fraction of the Compton electron spectrum of the 4.43-MeV gamma above T, and ρ is the density of the scintillator. Here ϵ_γ as a function of E was calculated with the assumption that the gamma angular distribution was independent of energy and normalizing to the total cross section as a function of energy. The contribution of the final-state neutron was calculated by use of Eq. (27) with $d\sigma^{n'}(n, n'\gamma)/d\Omega$ substituted for $d\sigma(n, n)/d\Omega$.

C. Results of Sample Calculations

The results of the calculation of the detection efficiency of a sample counter are shown in Fig. 3. The detector geometry was approximated as a right cylinder with the axis along the incident-neutron direction. The average escape distance was computed for uniform neutron illumination of the cross-sectional area by a separate program (see Appendix C). The results of this program were empirically fitted with a power series in $\cos\theta$. This power series was used as the basis for the required subroutine in the main program. The contributions of the individual processes are also indicated in Fig. 3.

The maximum uncertainty exists in the region in which the efficiency due to the (n, n'3α) and (n, p) reactions rises from zero to maximum, i. e., 20 to 75 MeV. At 90 MeV the measurements by Kellogg indicate that essentially all the nonelastic n-C interactions produce charged particles with sufficient light production to exceed the 2.3-MeV threshold.[7]

The efficiency of a similar type of neutron counter has been measured in the energy region 4 to 76 MeV.[12] The program computations and hand calculations described here were performed for this counter. The threshold was taken to be 1.1 MeV with a 50% resolution, and the geometry approximated by a cylinder with a thickness of 15 cm and a radius of 10 cm. The results of the calculation and the measured points are shown in Fig. 4. A measurement at 27-MeV neutron energy of the efficiency of a neutron detector with all parameters equal to those listed in the caption of Fig. 3.[13] This measurement was not corrected for rescattering and absorption losses in the target that served as the neutron source, and therefore would be expected to be slightly low in comparison with the calculated value.

The uncertainty in the calculated value of the efficiency was estimated to be ±10%. This number was selected because it is the order of uncertainty in the total nonelastic n-C cross section. The application of a χ^2 test of statistical significance to the comparison with the measurements by Wiegand et al.[12] yielded a result of $\chi^2/(\text{degrees of freedom}) \approx 1$, indicating the calculation to be compatible with the uncertainty quoted for the measured efficiency.

This work was performed under the auspices of the U. S. Atomic Energy Commission.



Fig. 3. Neutron-detection efficiency.

10.16-cm thick, 9.0-cm radius, cylindrical plastic scintillator
2.3 ± 0.7-MeV threshold (equivalent electron energy)

1: total efficiency
2: n-p scattering contribution
3: (n, n'3α) contribution
4: (n, p) contribution
5: (n, n'γ) contribution
6: n-p rescattering contribution
7: (n, n'3α) rescattering contribution
O: data from reference 13.



Fig. 4. Comparison of calculation with efficiency measured by Wiegand et al. (reference 12).

APPENDICES

A. Program Description and Summary of Input Data

System: 709/7090 FORTRAN II

Component Programs:

1. Main program: TOTEFF
2. Subroutines:
   a. HYD (E, F)
   b. CARB(E, F)
3. FORTRAN functions:
   a. SIG(E, N), which computes interpolated values of the cross section from input-cross-section data for the reaction designated by N at neutron energy E.

N = 1	:	C (n, α)Be
= 2	:	C (n, n'3α)
= 3	:	C (n, p)B
= 4	:	H (n, n)H
= 5	:	C (n, n')C total cross section

   b. D(X'), which computes the average distance to escape the detector in a direction corresponding to a c.m.-system scattering angle of $\cos^{-1} X'$ in an n-p collision.[a]
   c. V(X), which computes the average distance to escape the detector at a lab-system angle of $\cos^{-1} X$.[a]

Input Data:

1. Type 1 card: IN, EMN, EMX, DE, DH, SDB, EB, TH, KN

   Format (I2, 7F7.3, I2), where

   IN = number of steps in the Gaussian numerical integrations,
   EMN = minimum incident neutron energy (MeV),
   EMX = maximum incident neutron energy (MeV),
   DE = interval of incident neutron energy steps (MeV),
   DH = number of hydrogen (= carbon) nuclei per cm ($\times 10^{-24}$),
   SDB = fractional threshold energy resolution,
   EB = detection threshold (electron MeV),
   TH = detector thickness (cm), and
   KN = number of steps in the numerical-threshold-resolution folding integration (Eq. 1).

2. Type 2 cards: (A(I), S(I), I = 1, IN)

   Format (6F10.8), where

[a] The functions depend upon the specific detector geometry (see Appendix C).

A(I) = abscissas of the Gaussian quadrature formula for IN steps,[14] and

S(I) = weight coefficients for the Gaussian quadrature formula for IN steps.[14]

3. Type 3 cards: (AF(I), SF(I), I = 1, KN)

Format (6F10.8), where:

AF(I) = abscissas of the Gauss-Hermite quadrature formula for KN steps,[14] and

SF(I) = weight coefficients for the Gauss-Hermite quadrature formula for KN steps.[14]

4. Type 4 cards: CS(27, 5)

Format (9F8.4), where:

CS(I, N) = cross section (in cm^{-24}) for the reaction designated by N (see Appendix A.3.a.) at neutron energy E(I),

where

$E(I) = 10^{(I-1)/10}$ MeV.

The cross-section values corresponding to Figs. 1 and 2 are given in Table III. Note that for N = 5 two alternatives are given corresponding to the discussion following Eq. (26).

Table III. Cross-section values (in cm^{-24}) for CS(I, N)

I	E (MeV)	N = 1	2	3	4	5	
							n-C total-diffraction elastic
1	1.0				4.25	2.60	2.60
2	1.26				3.72	2.30	2.30
3	1.59				3.27	2.03	2.03
4	2.00				2.89	1.73	1.73
5	2.51				2.52	1.56	1.56
6	3.16				2.20	1.90	1.90
7	3.98				1.88	2.00	2.00
8	5.01	0.0	0.0		1.62	1.25	1.25
9	6.31	0.0	0.0		1.37	1.05	1.05
10	7.94	0.04	0.0	0.0	1.13	1.09	0.96
11	10.0	0.20	0.020	0.0	0.935	1.15	0.85
12	12.6	0.13	0.170	0.0	0.750	1.28	0.72
13	15.9	0.06	0.315	0.005	0.610	1.43	0.70
14	20.0	0.025	0.285	0.080	0.480	1.45	0.62
15	25.1	0.0	0.243	0.140	0.380	1.40	0.56
16	31.6	0.0	0.203	0.140	0.290	1.23	0.44
17	39.8		0.170	0.130	0.223	1.05	0.38
18	50.1		0.147	0.125	0.172	0.88	0.322
19	63.1		0.127	0.118	0.128	0.72	0.284
20	79.4		0.113	0.112	0.096	0.58	0.257
21	100.0		0.106	0.108	0.071	0.47	0.240
22	125.9		0.103	0.102	0.057	0.38	0.230
23	158.5		0.101	0.100	0.048	0.325	0.223
24	199.5		0.100	0.100	0.0415	0.295	0.217
25	251.1		0.100	0.100	0.037	0.285	0.212
26	316.2		0.100	0.100	0.034	0.285	0.205
27	398.1		0.100	0.100	0.0335	0.295	0.200

B. Source-Program Listings

```
   MAIN PROGRAM - TOTEFF

 C EQUATION NUMBERS REFER TO THE TEXT
       DIMENSION A(16),S(16),AF(16),SF(16),      CS(27,5),W(6),X(16),Y(16
      1),Z(16),EN(16,16),WG(16,16),EFF(6),FR(6),C(3),AL(16)
       COMMON CS,SCL,IN,JN,X,Y,Z,W,FR,C,XMAX,G
 C     ARITHMETIC STATEMENT FUNCTION DEFINITIONS
       TMAXF(X,Y) = ((ECM - X)**2 - Y**2)/(2.*ECM)
       NUMF(X,Y) = GAMMA*(X + Y)
       DENF(X,Y) = ETA*SQRTF(X*(X + 2.*Y))
       TLABF(X,Y) = 2.*X*(BETA**2*Y)/(1. - (BETA**2*Y))
 C READ INPUT DATA
 C READ TYPE 1 CARD
    10 READ INPUT TAPE 2,11,IN,EMN,EMX,DE,DH,SDB,EB,TH,KN
 C READ TYPE 2 CARDS
    12 READ INPUT TAPE 2, 9,(A(I),S(I),I=1,IN)
 C READ TYPE 3 CARDS
     8 READ INPUT TAPE 2,9,(AF(I),SF(I),I=1,KN)
 C READ TYPE 4 CARDS
    14 READ INPUT TAPE 2,15,CS
     9 FORMAT(6F10.8)
    11 FORMAT(I2,7F7.3,I2)
    15 FORMAT(9F8.4)
    18 WRITE OUTPUT TAPE 3,19,TH,SDB,EB,IN,EMN,EMX,DE
    19 FORMAT(1H1,39HPLASTIC SCINTILLATOR NEUTRON EFFICIENCY/
      130HOCARBON COLLISIONS CALCULATION,10X,31HHYDROGEN COLLISIONS CALCU
      2LATION/32HO(3 BODY PLUS NEUTRON RESCATTER),8X,37H(RELATIVISTIC PLU
      2S NEUTRON RESCATTER)/
      31HO,F6.2,11H CM. THICK,F8.2,19HLIGHT RESOLUTION AT,F6.2,5H MEV./
      419HOINTEGRATION STEP =I4/15HONEUTRON ENERGY,36X,17HCARBON EFFICIEN
      5CY28X,19HHYDROGEN EFFICIENCY/
      66H MIN =F8.2,11X,5HTOTAL,10X,5H(N,A),10X,5H(N,P),7X,8H(N,N*3A),
      76X,9H(N*RESCT),10X,5H(N,N),6X,9H(N*RESCT)/6H MAX =F8.2/7H STEP =
      8F7.2///)
 C SET CONSTANTS
       PI = 3.1415926
       SCL = 10./LOGF(10.)
       W(1) =  3727.22
       W(2) =   939.51
       W(3) =   938.21
       W(4) =  8392.40
       W(5) = 11174.40
       W(6) = 11188.28
 C NORMALIZATION FOR INTEGRATION OF EQUATION (1)
       AMB = 1.414214*EB*SDB
       XNM = 1./(1. + .2316418/SDB)
       XNM = 1. - ((((.9406607*XNM - 1.287822453)*XNM + 1.25969513)*XNM
      1- .252128668)*XNM + .225836846)*XNM*.56418959*EXPF(-.5/SDB**2)
 C SET UP NEUTRON ENERGY
   500 IE = (EMX - EMN)/DE + 1.0001
   501 E = EMN
```

MAIN PROGRAM - TOTEFF

```
  502 DO 542 I = 1,IE
C COMPUTE KINEMATICAL QUANTITIES
  503 ECM = SQRTF ((W(2) + W(5))**2 + 2.*E*W(5))
  504 ETA = SQRTF(E*(E + 2.*W(2)))/ECM
  505 GAMMA = (E + W(2) + W(5))/ECM
C SET UP QUANTITIES USED IN RESCATTER CALCULATIONS
  506 DO 512 J = 1,IN
  507 X(J) = (A(J) + 1.)*.5
  508 Z(J) = S(J)*SQRTF(X(J))*(1. - X(J))**2
  510 DO 512 L = 1,IN
  511 EN(J,L) = NUMF(X(J)*TMAXF(W(2),W(5)),W(2)) + A(L)*DENF(X(J)*TMAXF(
     1W(2),W(5)),W(2)) - W(2)
  512 WG(J,L) = (1. - EXPF(-DH*(SIG(5,EN(J,L))+SIG(4,EN(J,L)))*V(A(L))))
     1*S(L)/(SIG(5,EN(J,L)) + SIG(4,EN(J,L)))
  513 DO 514 J = 1,6
  514 EFF(J) = 0.
C SET UP INTEGRAL IN EQUATION (1)
  515 DO 533 K = 1,KN
  516 F = AMB*AF(K) + EB
  517 IF(F)533,533,518
C CARBON COLLISIONS
C SINGLE SCATTER N-CARBON
  518 CALL CARB(E,F)
      JN = JN
C EQUATION (1)
 5181 DO 5182 J = 1,JN
 5182 EFF(J) = EFF(J) + SF(K)*FR(J)
  519 GO TO (400,520,520),JN
  520 IF(FR(2) - .95)5201,5201,400
 5201 FR(4) = 0.
C RESCATTER N-CARBON
  300 DO 310 J = 1,IN
  301 AL(J) = F - .046*Y(J)
  302 Y2 = .007*Y(J)**2
C EQUATION (23)
  303 IF(AL(J) - Y2/3.)304,304,306
  304 AL(J) = 0.
  305 GO TO 310
C EQUATION (24)
  306 IF(AL(J) - Y2/2.)309,307,307
  307 AL(J) = AL(J) - 5./12.*Y2
  308 GO TO 310
C EQUATION (25)
  309 AL(J) = (AL(J) - Y2/3.)*.5
  310 CONTINUE
C EQUATION (21)
  521 DO 531 J = 1,IN
  523 G2 = 0.
  524 IF(AL(J))531,531,526
  526 DO 530 L = 1,IN
```

```
      MAIN PROGRAM - TOTEFF

     5261 IF(EN(J,L) - AL(J)    )530,530,527
      527 CALL CARB(EN(J,L),AL(J))
          JN = JN
          EC = 0.
      528 DO 529 M = 1,JN
      529 EC = SIG(M,EN(J,L))*FR(M) + EC
          CALL HYD(EN(J,L),AL(J))
          G2 = G2 + WG(J,L)*(.002*PI*G + EC)
      530 CONTINUE
      531 FR(4) = FR(4) + Z(J)*G2
      532 EFF(4) = EFF(4) + FR(4)*SF(K)/.30476
C     HYDROGEN COLLISIONS
C     SINGLE SCATTER N-HYDROGEN
      400 CALL HYD(E,F)
          BETA = SQRTF(E*(E + 2.*W(2)))/(E + W(2) + W(3))
C     EQUATION (1)
      401 EFF(5) = EFF(5) + SF(K)*G
      402 IF(G*.002*PI-.95*SIG(4,E))403,403,533
C     RESCATTER N-HYDROGEN
C     EQUATIONS (7) AND (8)
      403 C0 = C(1)
          C2 = C(2)
          C4 = C(3)
          FR(6) = 0.
          XP = (1. - XMAX)*.5
          XQ = (1. + XMAX)*.5
          DO 412 J = 1,IN
      405 XCM = XP*A(J) + XQ
      406 FP = .929*(TLABF(W(3),.5*(1. - XCM))) + 10.68*(1. - EXPF(-.07*(TLA
         1BF(W(3),.5*(1. - XCM)))**.89))
      600 IF(FP)601,602,602
      601 FP = 0.
      602 IF(F - FP)603,604,604
      603 FP = F
      604 ENP = TLABF(W(2),.5*(XCM + 1.))
      407 CALL HYD(ENP,F-FP)
      408 CALL CARB(ENP,F-FP)
          JN = JN
          EC = 0.
      409 DO 410 M = 1,JN
      410 EC = EC + SIG(M,ENP)*FR(M)
      411 FR(6) = FR(6) + S(J)*(1. - EXPF(-DH*(SIG(5,ENP) + SIG(4,ENP))*
         1D(XCM)))/(SIG(5,ENP)+ SIG(4,ENP))*(EC + .002*PI*G)*(C0 + C2*XCM**2
         2 + C4*XCM**4)
      412 CONTINUE
      413 EFF(6) = EFF(6) + SF(K)*XP*FR(6)
      533 CONTINUE
C     NORMALIZATION OF NUMERICAL INTEGRALS AND PRINTOUT
      534 CRB = SIG(4,E) + SIG(5,E)
      535 CRB = (1. - EXPF(-DH*CRB*TH))/(CRB*SDB*EB*2.507)*AMB/XNM
```

MAIN PROGRAM - TOTEFF

```
5361 EFF(4) = EFF(4)*CRB*SIG(2,E)
     EFF(5) = EFF(5)*CRB*.002*PI
     EFF(6) = EFF(6)*CRB*.002*PI
 536 EF = 0.
 537 DO 539 J = 1,3
 538 EFF(J) = EFF(J)*CRB*SIG(J,E)
 539 EF = EF + EFF(J) + EFF(J + 3)
 540 WRITE OUTPUT TAPE 3,541,E,EF,EFF(1),EFF(3),EFF(2),EFF(4),EFF(5),
    1EFF(6)
 541 FORMAT(1HO,8X,F6.2,7F15.8)
 542 E = E + DE
 543 GO TO 10
     END(1,0,0,0,0,0,1,0,0,1,0,0,0,0,0)
```

```
      SUBROUTINE HYD(E,F)

      SUBROUTINE HYD(E,F)
      DIMENSION W(6),C(3),X(16),Y(16),Z(16),FR(6),CS(27,5)
      COMMON CS,SCL,IN,JN,X,Y,Z,W,FR,C,XMAX,G
      XMAX = -1.
      G = 0.
      COR = 1.0
      DO 99 I = 1,3
   99 C(I) = 0.
C CALCULATE KINEMATICS
  100 ECM = SQRTF((W(2) + W(3))**2 + 2.*W(3)*E)
  101 ETA = SQRTF(E*(E + 2.*W(2)))/ECM
  102 GAMMA = (E + W(2) + W(3))/ECM
  103 BETA = ETA/GAMMA
C EQUATION (6)
  104 ELIM = 1.077*F + 11.5*(1. - EXPF(-.2*F**.62)) + W(3)
  105 IF(ELIM - E - W(3))106,158,158
  106 XMAX = (GAMMA*(ECM**2 + W(3)**2 - W(2)**2)/(2.*ECM) - ELIM)/(ETA*
     1SQRTF(((ECM**2 + W(3)**2 - W(2)**2)/(2.*ECM))**2 - W(3)**2))
  107 IF(XMAX - 1.)109,109,108
  108 XMAX = 1.
C EQUATION (5) USING CLEMENTS AND WINSBURG
C SELECT ENERGY RANGE
  109 E1 = E/W(2)
      D1 = 1./E1
      E2 = E1**2
      D2 = 1./E2
      E3 = E1**3
      D3 = 1./E3
      E4 = E2**2
      IF(E1- .0002)115,120,110
  110 IF(E1-.015)120,125,111
  111 IF(E1- .2)125,140,112
  112 IF(E1- .62)140,155,155
C ENERGY RANGE (0 - .0002)
  115 C(1) = 1.62E+3 - 7.92E+6*E1+ 1.89E+10*E2
      C(2) = 0.
      C(3) = 0.
  116 G = (XMAX + 1.)*C(1)
      GO TO 158
C ENERGY RANGE (.0002 - .015)
  120 C(1) = -3400.*E1+ 78.93 + .3757*D1- 9.977E-5*D2    + 1.0608E-9*D3
      IF(E1-.003)121,121,163
  163 COR = 2180.*(E1-.009)**2 + .96
  121 C(2) = 0.
      C(3) = 0.
      GO TO 116
C ENERGY RANGE (.015 - .200)
  125 CT = 2.1 - .1624*D1+ .04011*D2    - 4.0E-4*D3
      IF(E1-.075)160,160,161
  160 COR = 2.92*E1 + .89
```

```
      SUBROUTINE HYD(E,F)

      GO TO 126
  161 COR = 60.*(E1-.125)**2 + .96
  126 CU = 1.57 + 251.7*E1- 5021.*E2    + 41192.*E3    - 1.162E+5*E4
  127 IF(CU)129,129,129
  128 CU = 0.
  129 CV = -41.*E1+ 2974.*E2    - 33053.*E3    + 108909.*E4
  130 IF(CV)131,136,136
  131 CV = 0.
  136 C(1) = -.9 + .377*D1+ .0134*D2    - 1.77E-4*D3
      C(2) = 0.
  137 C(3) = 182.9*E1- 1405.8*E2    + 3066.2*E3
  132 G = CU/3. + CV/15. + CT
  133 IF(XMAX)134,159,138
  134 G = G + CT*XMAX + CU/3.*XMAX**3 + CV/15.*XMAX**15
  135 GO TO 158
  138 G = G + C(1)*XMAX + C(3)/5.*XMAX**5
  139 GO TO 158
C ENERGY RANGE (.200 - .620)
  140 CW = -.5 + 15.26*E1- 41.64*E2    + 25.08*E3
      IF(E1-.425)164,164,141
  164 COR = 7.65*(E1-.309)**2 + .96
  141 CX = 37.42 - 227.38*E1+ 512.62*E2    - 380.37*E3
  142 CY = -4.42 + 35.43*E1- 78.62*E2    + 59.52*E3
  143 CZ = 4.
  148 C(1) = 1.32 + 3.95*E1- 6.71*E2
  149 C(2) = 4.19 + 3.34*E1- 44.86*E2    + 62.12*E3
      C(3) = 0.
  144 G = CW + CX/5. + CY/15. + CZ/101.
  145 IF(XMAX)146,158,150
  146 G = G + CW*XMAX+CX/5.*XMAX**5+CY/15.*XMAX**15+CZ/101.*XMAX**101
  147 GO TO 158
  150 G = G + C(1)*XMAX + C(2)/3.*XMAX**3
  151 GO TO 158
C ENERGY RANGE (.620 - 1.065)
  155 C(1) = -20.84 + 88.88*E1- 114.82*E2    + 47.08*E3
      C(2) = 0.
  156 C(3) = 149.74 - 562.32*E1+ 704.13*E2    - 285.13*E3
  157 G = C(1)*(XMAX + 1.) + C(3)/5.*(XMAX**5 + 1.)
  158 G = G*COR
      DO 159 I = 1,3
  159 C(I) = COR*C(I)
      RETURN
      END(1,0,0,0,0,0,0,0,0,1,0,0,0,0,0)
```

```
      SUBROUTINE CARB(E,F)

      SUBROUTINE CARB(E,F)
      DIMENSION T(3),FR(6),TL(3),GG(3),W(6),X(16),Y(16),Z(16),CS(27,5),
     1C(3)
      COMMON CS,SCL,IN,JN,X,Y,Z,W,FR,C,XMAX
C     ARITHMETIC STATEMENT FUNCTION DEFINITIONS
      TMAXF(X,Y) = ((ECM - X)**2 - Y**2)/(2.*ECM)
      NUMF(X,Y) = GAMMA*(X + Y)
      DENF(X,Y) = ETA*SQRTF(X*(X + 2.*Y))
  100 DO 101 J= 1,3
  101 FR(J) = 0.
      IN = IN
  102 JN = 1
C CHECK THRESHOLDS AND CALCULATE KINEMATICS
  103 IF(E - 6.5)132,104,104
  104 ECM = SQRTF((W(2) + W(5))**2 + 2.*E*W(5))
  105 ETA = SQRTF(E*(E + 2.*W(2)))/ECM
 1050 GAMMA = (E + W(2) + W(5))/ECM
  106 IF(E - 8.) 114,107,107
  107 IF(E - 14.)110,112,112
  110 JN = 2
  111 GO TO 114
  112 JN = 3
  113 T(3) = TMAXF(W(3),W(6))
C EQUATION (3)
 1131 TL(3) = 1.077*F + 11.5*(1. - EXPF(-.2*F**.62)) + W(3)
  114 T(1) = TMAXF(W(1),W(4))
C EQUATION (3)
 1141 TL(1) = 1.5*F + 50.*(1. - EXPF(-.16*F**.62)) + W(1)
C EQUATION (11)
 1142 FR(1) = .5 + (NUMF(T(1),W(1)) - TL(1))/(DENF(T(1),W(1))*2.)
 1143 IF(FR(1) - 1.)1145,1147,1144
 1144 FR(1) = 1.
 1145 IF (FR(1))1146,1147,1147
 1146 FR(1) = 0.
 1147 L = 3
  115 DO 129 J = 1,IN
  116 GO TO (132,123,117),JJ
C EQUATION (18)
  117 GG(L) = (TL(L) - NUMF(T(L)*X(J),W(L)))/DENF(T(L)*X(J),W(L))
C EQUATION (17)
  118 IF(ABSF(GG(L)) - 1.)120,120,119
  119 GG(L) = SIGNF(1.,GG(L))
  120 GG(L) = (1. - GG(L))*.5
C EQUATION (19)
  121 FR(L) = FR(L) + Z(J)*GG(L)
C EQUATION (13)
  123 Y(J) = -(W(5) + 7.37) + SQRTF(X(J)*(W(5) + 7.37)**2 + (1. - X(J))*
     1(ECM - W(2))**2)
C EQUATION (15)
  124 GG(2) = 3.*(1. - (F/Y(J) - .046)/(.0035*Y(J)))
```

```
      SUBROUTINE CARB(E,F)

  125 IF(GG(2))129,129,126
  126 IF(GG(2) - 1.)128,128,127
  127 GG(2) = 1.
C EQUATION (16)
  128 FR(2) = FR(2) + Z(J)*GG(2)
  129 CONTINUE
C NORMALIZATION
  130 DO 131 J = 2,JN
  131 FR(J) = FR(J)/.30476
  132 RETURN
      END(1,0,0,0,0,0,0,0,0,1,0,0,0,0,0)
```

```
      FUNCTION SIG(N,E)

      FUNCTION SIG(N,E)
      DIMENSION CS(27,5)
      COMMON CS,SCL
      IF(E - 1.)605,605,600
  600 Z = SCL*LOGF(E)      + 1.
  601 L = Z
  603 SIG = CS(L,N) + MODF(Z,1.)*(CS(L+1,N) - CS(L,N))
  604 RETURN
  605 GO TO (606,606,606,608,610),N
  606 SIG = 0.
  607 RETURN
  608 SIG = 20./(1. + 3.7*E)
  609 RETURN
  610 SIG = 4.7 - 2.1*E
  611 RETURN
      END(1,0,0,0,0,0,0,0,0,1,0,0,0,0,0)
```

C. Sample Treatment of Detector Geometry

The 709/7090 FORTRAN II program listed below was used to compute the average escape distance from a cylinder as a function of angle with respect to the axis. The results were empirically fitted to a power series in the cosine of the angle and this function used for the required FORTRAN functions, D(X') and V(X), which are also listed below.

```
      PROGRAM TO COMPUTE THE AVERAGE  ESCAPE DISTANCE FROM A CYLINDRICAL
C DETECTOR UNIFORMLY ILLUMINATED BY NEUTRONS INCIDENT IN THE DIRECTION
C OF THE AXIS
C TL = CYLINDER THICKNESS (CM.)
C TR = CYLINDER RADIUS (CM.)
C THX = MAXIMUM ANGLE (DEG)
C THN = MINIMUM ANGLE (DEG)
C DTH = SIZE OF ANGULAR STEPS (DEG)
C IN = NUMBER OF STEPS IN GAUSSIAN QUADRATURES
C A(I) = ABSCISSAE OF GAUSSIAN QUADRATURE FORMULA
C W(I) = WEIGHTING COEFFICIENTS FOR GAUSSIAN QUADRATURE FORMULA
      DIMENSION A(16),W(16),Z(16),R(16),CR(16),BR(16),B(2,16,16)
    1 READ INPUT TAPE 2,2,TL,TR     ,THX,THN,DTH,IN
    2 FORMAT(3F10.5,I2)
    3 READ INPUT TAPE 2,4,(A(I),W(I),I=1,IN)
    4 FORMAT( 6F10.8)
    5 WRITE OUTPUT TAPE 3,6,TL,TR,IN
    6 FORMAT(1H1,46HAVERAGE DISTANCE TO ESCAPE CYLINDRICAL COUNTER/
     117HOCOUNTER LENGTH =F10.5/17H COUNTER RADIUS =F10.5/
     220H INTEGRATION STEPS =I4/6HOTHETA,6X,9HAVE. DIST//)
      PI = 3.141592654
   10 DO 14 I = 1,IN
   11 Z(I) = TL/2.*(A(I) + 1.)
   12 R(I) = TR/2.*(A(I) + 1.)
   13 CR(I) = COSF(PI/4.*(A(I) + 1.))
   14 BR(I) = (TR/R(I))**2 - 1.
   15 DO 19 I = 1,IN
   16 DO 19 J = 1,IN
   17 DR = SQRTF(BR(I) + CR(J)**2)
   18 B(1,I,J) = R(I)*(DR - CR(J))
   19 B(2,I,J) = R(I)*(DR + CR(J))
   20 ITH = (THX - THN)/DTH + 1.01
   21 TH = THN
   22 DO 38 I = 1,ITH
   23 CTH = COSF(TH*PI/180.)
   24 STH = SINF(TH*PI/180.)
   25 RAV = 0.
   26 DO 34 J = 1,IN
   27 PHAV = 0.
   28 DO 33 K = 1,IN
   29 ZAV = 0.
   30 DO 32 L = 1,IN
   32 ZAV = ZAV + W(L)*(MINIF(B(1,J,K)/STH,Z(L)/CTH) + MINIF(B(2,J,K)/
     1STH,Z(L)/CTH))
   33 PHAV = PHAV + W(K)*ZAV
   34 RAV = RAV + R(J)*PHAV*W(J)
   35 RAV = RAV/(8.*TR)
   36 WRITE OUTPUT TAPE 3,37,TH,RAV
   37 FORMAT(F8.2,F15.5)
   38 TH = TH + DTH
```

```
PROGRAM TO COMPUTE THE AVERAGE  ESCAPE DISTANCE FROM A CYLINDRICAL
 39 GO TO 1
    END(1,0,0,0,0,0,0,0,0,1,0,0,0,0,0)
```

```
FUNCTION D(X)

FUNCTION D(X)
Y = SQRTF((X + 1.)*.5)
D = 13.2 + 10.0*Y**4 + 3.0*Y**10
RETURN
END(1,0,0,0,0,0,0,0,0,1,0,0,0,0,0)
```

```
FUNCTION V(X)

FUNCTION V(X)
Y = ABSF(X)
V = 13.2 + 10.0*Y**4 + 3.0*Y**10
RETURN
END(1,0,0,0,0,0,0,0,0,1,0,0,0,0,0)
```

REFERENCES

1. H. Grassler and K. Tesch, Nucl. Instr. Methods 10, 353 (1961); James E. Hardy, Rev. Sci. Instr. 29, 705 (1958); Lawrence Cranberg and Jules S. Levin, Phys. Rev. 103, 343 (1956); R. Batchelor, W. B. Gilboy, J. B. Parker, and J. H. Towle, Nucl. Instr. Methods 13, 70 (1961).

2. Donald J. Hughes and Robert B. Schwartz, Neutron Cross Sections, Brookhaven National Laboratory Report BNL-325, July 1958 (unpublished).

3. Wilmot N. Hess, Rev. Mod. Phys. 30, 368 (1958).

4. Malcolm H. MacGregor, William P. Ball, and Rex Booth, Phys. Rev. 111, 1155 (1958); G. L. Salmon, Nucl. Phys. 21, 15 (1960); A. Ashmore, R. G. Jarvis, D. S. Mather, and S. K. Sen, Proc. Phys. Soc. (London) 70A, 745 (1957); A. Ashmore, D. S. Mather, and S. K. Sen, Proc. Phys. Soc. (London) 71, 552 (1957); P. H. Bowen, J. P. Scanlon, G. H. Stafford, J. J. Thresher, and P. E. Hodgson, Nucl. Phys. 22, 640 (1961); C. P. van Zyl, R. G. P. Voss, and R. Wilson, Phil. Mag. 1, 1003 (1956); A. E. Taylor, T. G. Pickavance, J. M. Cassels, and T. C. Randle, Phil. Mag. 42, 328 (1951); A. E. Taylor and E. Wood, Phil. Mag. 44, 95 (1952); R. Fox, C. Leith, L. Wouters, and K. R. MacKenzie, Phys. Rev. 80, 23 (1950); James DeJuren, Phys. Rev. 80, 27 (1950); J. DeJuren and B. J. Moyer, Phys. Rev. 81, 919 (1950); George R. Mott, Gordon L. Guernsey, and Bruce K. Nelson, Phys. Rev. 88, 9 (1952); A. E. Taylor, Phys. Rev. 92, 1071 (1953); V. Alexander Nedzel, Phys. Rev. 94, 174 (1954).

5. M. P. Nakada, J. D. Anderson, C. C. Gardner, and C. Wong, Phys. Rev. 110, 1439 (1958); J. D. Anderson, C. C. Gardner, J. W. McClure, M. P. Nakada, and C. Wong, Phys. Rev. 111, 572 (1958).

6. G. M. Frye, Jr., L. Rosen, and L. Stewart, Phys. Rev. 99, 1375 (1955).

7. D. A. Kellogg, Phys. Rev. 90, 224 (1953).

8. Thomas P. Clements and Lester Winsberg, Polynomial Fits of Nucleon-Nucleon Scattering Data, UCRL-9043, Feb. 1960.

9. T. J. Gooding and H. G. Pugh, Nucl. Instr. Methods 7, 189 (1960).

10. F. Ajzenberg-Selove and T. Lauritsen, Energy Levels of Light Nuclei, Ann. Rev. Nucl. Sci. 10, 416 (1960).

11. Hans A. Bethe and Julius Ashkin, Passage of Radiation Through Matter, in Experimental Nuclear Physics, (John Wiley and Sons, Inc., New York, 1953), pp. 317-325.

12. Clyde E. Wiegand, Tom Elioff, William B. Johnson, Leonard B. Auerbach, Joseph Lach, and Thomas Ypsilantis, Detection Efficiency of Plastic Scintillator for Neutron Energies 4 to 76 MeV, UCRL-9986, Dec. 1961.

13. G. Gatti, P. Hillman, W. C. Middlekoop, T. Yamagata, and E. Zavattini, Phys. Rev. Letters 6, 706 (1961).

14. Zdenek Kopal, Numerical Analysis, 2d. ed., (John Wiley and Sons, Inc., New York, 1961), pp. 367-373, 523-533.